A robust, sensitive, and versatile HMBC experiment for rapid structure elucidation by NMR: IMPACT-HMBC

Julien Furrer

Supplementary information

1. Structures and atom numbering of the molecules used in this work



Fig 1a Chloesteryl acetate



Fig 1b Isopropylidene glycerol



Fig 1c Cyclosporine

2. Standard HMBC pulse sequence used in this work



Fig 2 Pulse sequence of the low-pass gradient-selected HMBC experiment used in this work, derived from the original HMBC (Bax & Summer, 1986). Narrow solid bars indicate 90° pulses and wide bars 180° pulses. $^{1}J_{CH}$ correlations are suppressed using a one-step low pass J filter, and the delay τ is set as an average $^{1}J_{CH}$ value, typically 145 Hz. The delay Δ is for the evolution of $^{n}J_{CH}$. Gradient ratios: G_1:G_2:G_3:G_4 = 5:3:4:2. The phases of the pulses are ϕ_1 = x, -x; ϕ_2 = x, x, -x,-x; ϕ_{REC} = x, -x, -x, x and x when not specified.

3. IMPACT-HMBC pulse programs for Bruker Avance Spectrometers

a. 2D version

```
;IMPACT-HMBC
;avance-version
;phase sensitive HMBC using Echo/Antiecho gradient selection
;with three-fold low-pass J-filter to suppress one-bond correlations
;no decoupling during acquisition
;using constant time for suppressing F1 skew
;using ASAP mixing for suppressing F1 ridges

; J. Furrer, Chem. Commun. 2010.

;Echo/Antiecho scheme
;D.O. Cicero, G. Barbato & R. Bazzo, J. Magn. Reson. 148, 209-213 (2001)

;Constant time
;K. Furihata & H. Seto, Tetrahedron Lett. 39, 7337-7340 (1998)

;ASAP method
;E. Kupce & R . Freeman, Magn. Reson. Chem. 45, 2-6 (2007)

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>
```

```
"cnst30=(1-sfo2/sfo1)/(1+sfo2/sfo1)"

define list<gradient> EA1 = { 1.000 -cnst30}
define list<gradient> EA2 = { -cnst30 1.000}

"p2=p1*2"
"d0=3u"
"in20=in0"

;"---------- Fixed delays, do not change ----------"

"DELTA1=1s/(2*(cnst6 + 0.07 * (cnst7-cnst6))) -p16 -d16"
"DELTA2=1s/(cnst7+cnst6) -p16 -d16"
"DELTA3=1s/(2*(cnst7 - 0.07 * (cnst7-cnst6))) -p16 -d16"
"DELTA4=1s/(cnst13*2)"
"DELTA5=p2+d0*2"
"FACTOR1=(d9/(p6*115.112))/2+0.5"
"l1=FACTOR1*2"
"d20=in20*td1/2+4u"


1 ze
2 d11
3 d1*0.5 UNBLKGRAD       ; beginning of the ASAP period
4 p16:gp6
  d16
  d12 pl10:f1
  p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
  p6*0.333 ph23
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23

  p6*3.556 ph25
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23
  p6*4.111 ph25

  p6*3.556 ph25
```

```
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23
  p6*4.111 ph25

  p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
  p6*0.333 ph23
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23
  lo to 4 times l1
  d12
  p16:gp6
  d16
  d1*0.5 pl1:f1          ; end of the ASAP period

5 p0 ph1                                ; Ernst angle
  DELTA1                                ; tau 1
  p16:gp2
  d16

  p3:f2 ph3
  DELTA2                                ; tau 2
  p16:gp3
  d16

  p3:f2 ph3
  DELTA3                                ;tau 3
  p16:gp4
  d16

  p3:f2 ph3
  DELTA4                                ;Delta
  p16:gp5
  d16

  d20                                   ;constant time delay

  p3:f2 ph4
  d0
```

```
  p2 ph2
  d0
  p16:gp1*EA1
  d16
  (p24:sp7 ph5):f2
  DELTA5
  p16:gp1*EA2
  d16 pl2:f2
  (p3 ph5):f2
  d20                                          ;constant time delay
  4u BLKGRAD
  go=2 ph31
  d11 mc #0 to 2
    F1EA(igrad EA1 & igrad EA2, id0 & dd20 & ip4*2 & ip31*2)
exit


ph1=0
ph2=0 0 0 0 2 2 2 2
ph3=0
ph4=0 2
ph5=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph6=0
ph7=1
ph23=1
ph25=3
ph31=0 2 0 2 0 2 0 2 2 0 2 0 2 0 2 0


;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl10 : f1 channel - power level for ASAP
;sp7: f2 channel - shaped pulse (180degree refocussing)
;spnam7: Crp60comp.4 for broadband experiment
;            : G3 (for band selective) or Gaussian (for selective)
;p0 : f1 channel -  high power pulse - Ernst Angle
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p6 : f1 channel -  90 degree ASAP pulse (ca. 50 us)
;p16: homospoil/gradient pulse                      [1 msec]
;p24: f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;d0 : incremented delay (2D)                        [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d9 : ASAP mixing time (ca. 40 ms)
;d16: delay for homospoil/gradient recovery
;d20: constant time delay (in20*td1/2+4u)
```

```
;cnst6: = 1J(XH)min
;cnst7: = 1J(XH)max
;cnst13: = J(XH) long range
;in0: 1/(2 * SW(X)) = DW(X)
;in20 : = in0 (decrementation step for d20)
;nd0: 2
;NS: 2 * n
;DS: 16
;td1: number of experiments
;FnMODE: echo-antiecho

;gpz1: 80%
;gpz2: 37%
;gpz3: -19%
;gpz4: -11%
;gpz5: -7%
;gpz6: -17%

;gpnam1-6: SINE.100

;PH_mod(F1): pk (or no)

;use xfb and xf2m
```

b. 1D version (for Ernst angle determination)

```
;IMPACT-HMBC-1d
;avance-version
;1D version of the IMPACT-HMBC, for determining the Ernst angle
;use popt (or paropt) and increment p0 from 10 deg to 180 deg in step of 10 deg

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"cnst30=(1-sfo2/sfo1)/(1+sfo2/sfo1)"

define list<gradient> EA1 = { 1.000 -cnst30}
define list<gradient> EA2 = { -cnst30 1.000}

"p2=p1*2"

;"------------ Fixed delays, Do not change --------------"

"DELTA1=1s/(2*(cnst6 + 0.07 * (cnst7-cnst6))) -p16 -d16"
"DELTA2=1s/(cnst7+cnst6) -p16 -d16"
```

```
"DELTA3=1s/(2*(cnst7 - 0.07 * (cnst7-cnst6))) -p16 -d16"
"DELTA4=1s/(cnst13*2)"
"DELTA5=p2+d0*2"
"FACTOR1=(d9/(p6*115.112))/2+0.5"
"l1=FACTOR1*2"
"d20=in20*td1/2+4u"


1 ze
2 d11
3 d1*0.5 UNBLKGRAD       ; beginning of the ASAP period
4 p16:gp6
  d16
  d12 pl10:f1
  p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
  p6*0.333 ph23
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23

  p6*3.556 ph25
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23
  p6*4.111 ph25

  p6*3.556 ph25
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23
  p6*4.111 ph25

  p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
```

```
  p6*0.333 ph23
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23
  lo to 4 times l1
  d12
  p16:gp6
  d16
  d1*0.5 pl1:f1              ;end of the ASAP period

5 p0 ph1                     ;Ernst angle
  DELTA1                     ;tau 1
  p16:gp2
  d16

  p3:f2 ph3
  DELTA2                     ;tau 2
  p16:gp3
  d16

  p3:f2 ph3
  DELTA3                     ;tau 3
  p16:gp4
  d16

  p3:f2 ph3
  DELTA4                     ;Delta
  p16:gp5
  d16

  d20                        ;constant time delay

  p3:f2 ph4
  3u
  p2 ph2
  3u
  p16:gp1*EA1
  d16
  (p24:sp7 ph5):f2
  DELTA5
  p16:gp1*EA2
  d16 pl2:f2
  (p3 ph5):f2
  d20                        ;constant time delay
  4u BLKGRAD
  go=2 ph31
d11 mc #0 to 2 F0(zd)
```

```
exit


ph1=0
ph2=0 0 0 0 2 2 2 2
ph3=0
ph4=0 2
ph5=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph6=0
ph7=1
ph23=1
ph25=3
ph31=0 2 0 2 0 2 0 2 2 0 2 0 2 0 2 0


;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl10 : f1 channel - power level for ASAP
;sp7: f2 channel - shaped pulse (180degree refocussing)
;spnam7: Crp60comp.4
;p0 : f1 channel -  high power pulse - Ernst Angle
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p6 : f1 channel -  90 degree ASAP pulse (ca. 50 us)
;p16: homospoil/gradient pulse                  [1 msec]
;p24: f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d9 : ASAP mixing time (ca. 40 ms)
;d16: delay for homospoil/gradient recovery
;d20: constant time delay (in20*td1/2+4u)
;cnst6: = 1J(XH)min
;cnst7: = 1J(XH)max
;cnst13: = J(XH) long range
;NS: 2 * n
;DS: 16

;gpz1: 80%
;gpz2: 37%
;gpz3: -19%
;gpz4: -11%
;gpz5: -7%
;gpz6: -17%

;gpnam1-6: SINE.100
```

4. Experimental details

The standard 5mm test tubes of cholesteryl acetate (100 mg/ml) dissolved in $CDCl_3$, cyclosporine dissolved in C_6D_6 (50 mmol/l) and a solution of ca. 20 mg of isopropylidene glycerol dissolved in 0.6 ml $CDCl_3$ were used for recording all experiments. All data were collected on a Bruker AvanceII 400 equipped with a BBFOplus double resonance probe (z-gradients) with the sample temperature regulated at 298 K. The ^{1}H 90° pulse length was 9.6 µs and the ^{13}C 90° pulse length 9.3 µs. Pulsed field gradients were applied as 1 ms sine shaped pulses. Shifted (SSB 2) qsine weighting was applied in both dimensions prior to Fourier transformation and all spectra are presented in magnitude-mode. For processing IMPACT-HMBC data, note that the data must be first processed in a phase-sensitive mode, and magnitude mode must be subsequently applied only in F_2. No further enhancement through linear prediction has been employed so that the effects of the sequence modifications remain clear.

Spectrum 2a was collected with the standard, non-selective pulsed field gradient HMBC sequence shown above over 5.5 ppm (^{1}H) × 180 ppm (^{13}C) as 2K × 256 data points, 4 transients, corresponding to a t_1(max) of 12 ms and an acquisition time of 0.25 s. The $\Delta\ ^nJ_{CH}$ coupling evolution delay was set to 62.5 ms and the $\tau\ ^1J_{CH}$ evolution delay to 3.45 ms. A recovery delay of 1 s was used, leading to a total experimental time of approximately 22 min. Data were zero-filled to 2K × 1K points.

Spectrum 2b was collected with the IMPACT-HMBC pulse sequence, over 5.5 ppm (^{1}H) × 180 ppm (^{13}C) as 2K × 256 data points, 8 transients, corresponding to a t_1(max) of 12 ms and an acquisition time of 0.25 s. The $\Delta\ ^nJ_{CH}$ coupling evolution delay was set to 62.5 ms and $^1J_{CHmax}$ and $^1J_{CHmin}$ were set to 170 and 120 Hz, respectively, leading to values of 4.04 ms for τ_1, of 3.45 ms for τ_2, and of 3 ms for τ_3. For the ASAP period, the DIPSI-2 mixing scheme was used. A 5 kHz rf field (50 µs pulse) was used and the mixing time was set to 40 ms. Alternatively, an adiabatic TOCSY mixing period, using a 50 µs TanhTan pulse, 100 kHz sweep, can be used. No significant difference has been observed between the two mixing schemes. A recovery delay of 0.2 s was used, leading to a total experimental time of approximately 22 min. Data were zero-filled to 2K × 1K points.

Spectrum 3a and 3b were collected over 5 ppm (^{1}H) × 120 ppm as 1K × 128 data points, 4 transients, an acquisition time of 0.25 s, leading to a total experimental time of approximately 2 min. For spectrum 3a, the $\Delta\ ^nJ_{CH}$ coupling evolution delay was set to 62.5 ms and the $\tau\ ^1J_{CH}$ evolution delay to 3.45 ms. For spectrum 3b, the $\Delta\ ^nJ_{CH}$ coupling evolution delay was set to 62.5 ms and $^1J_{CHmax}$ and $^1J_{CHmin}$ were set to 170 and 120 Hz, respectively. A 40 ms ASAP period was used, using a DIPSI-2 mixing scheme (5 kHz rf field (50 µs pulse)).

F_2 rows shown in figure 4 were extracted from the IMPACT-HMBC spectra, recorded over 5.5 ppm (^{1}H) × 180 ppm (^{13}C) as 2K × 256 data points, 8 transients, and an acquisition time of 0.25 s. The $\Delta\ ^nJ_{CH}$ coupling evolution delay was set to 62.5 ms and $^1J_{CHmax}$ and $^1J_{CHmin}$ were set to 170 and 120 Hz, respectively. An ASAP period 40 ms was used, using a 5 kHz rf field (50 µs pulse).

5. Band-selective ability

The versatility of the IMPACT-HMBC experiment is readily shown in Fig. 3. Fig. 3a shows the full width HMBC spectrum, recorded using the standard HMBC pulse sequence and using a 200 ppm ^{13}C window. Fig. 3b shows a semi-selective spectrum, using the IMPACT-HMBC pulse sequence (Fig. 1, main text), in which the broadband 180° pulse on the ^{13}C channel has been replaced by a 50 ppm band selective pulse. As shown by Fig. 3b, the IMPACT-HMBC sequence combines signals with no coupling structure, optimal $^{1}J_{CH}$ suppression, and very good signal to noise ratio using a minimum experimental time. Carbons with very similar chemical shifts, (indicated by horizontal arrows) can be easily resolved using the semi-selective version.



Fig 3 Excerpts of two-dimensional HMBC spectra of cholesteryl acetate recorded (a) with the standard HMBC experiment, full width spectrum, and (b) with the IMPACT-HMBC experiment, band-selective experiment, using a 500 μs G3 pulse that covers a 50 ppm ^{13}C window. (a) recorded with 128 increments, 4 transients, and a recovery delay of 1s. (b) recorded with 256 increments, 4 transients, and a recovery delay of 0.2s. The flip angle α was set to 90°. Measurement duration for both experiments 11 min. Dashed ellipses indicate residual $^{1}J_{CH}$ signals. Note the optimal $^{1}J_{CH}$ suppression and the absence of F_1 ridges in the IMPACT-HMBC spectrum. Note also that carbons C-19 and C-21 (indicated by horizontal arrows) have very similar chemical shifts. They can be easily resolved in the IMPACT-HMBC spectrum.

6. Ernst angle determination

The Ernst angle is determined rapidly (~5 min) with a one dimensional version of the IMPACT-HMBC pulse sequence. Spectra shown in figure 4 were collected over 5.5 ppm using the above 1D pulseprogram, as 2K data points, a recovery delay time of 200 ms, 4 transients, by incrementing the first proton hard pulse (p0) from 1.07 µs (10°) to 19.2 µs (180°) in step of 1.07 µs (10°) using the *popt* AU program. Shifted (SSB 2) qsine weighting was applied prior to Fourier transformation.



Fig 4 Ernst angle determination for cholesteryl acetate. The plot shows the behavior of the proton spectrum of cholesteryl acetate with increasing flip-angle of the initial hard pulse. The maximum signal amplitude is attained for an Ernst angle of about 110°, or ca. 11.7 µs (indicated by a vertical arrow).

7. Ernst angle determination for cyclosporine and isopropylidene glycerol

The following spectra have been recorded on cyclosporine (1.2 kDa), with nearly identical short (<1 s) T_1 and on isopropylidene glycerol (132 Da) with long (>3 s) T_1. Depending on the acquisition parameters (recovery delay and acquisition time), higher gain in the sensitivity can be achieved in an experiment using optimized Ernst angle excitation versus 90° pulsing.

cyclosporine



Fig 5 Ernst angle determination for cyclosporine. The plot shows the behavior of the proton spectrum with increasing flip-angle of the initial hard pulse. Spectra were collected over 10 ppm using the above 1D pulseprogram, as 2K data points, and using 4 transients. It can be seen that the maximum signal amplitude is attained for an Ernst angle between 90° and 100°, whatever the recovery delay value.

Isopropylidene glycerol



Fig 6 Ernst angle determination for isopropylidene glycerol. The plot shows the behavior of the proton spectrum with increasing flip-angle of the initial hard pulse. Spectra were collected

over 5 ppm using the above 1D pulseprogram, as 1K data points, and using 4 transients. It can be seen that the maximum signal amplitude is attained for an Ernst angle of 110° for a recovery delay of 1 s, of 120° for a recovery delay of 0.5 s, and of 130° for a recovery delay of 0.2 s. A gain of approximately 50% is realized for the last case using optimized Ernst angle versus 90° excitation.

It turns out from the above spectra that for molecules having long T_1 (>3 s), specifically small molecules or macromolecules, a significant gain in sensitivity compared to the standard setup using 1 s repetition and 90° excitation can be obtained by optimizing both the Ernst angle and the repetition rate. If the interscan delay is kept short (<1 s), a much higher gain in the relative sensitivity is achieved in an experiment using optimized Ernst angle excitation versus 90° pulsing. For medium-size molecules having short T_1 (<1 s), the gain in sensitivity compared to the standard setup using 1 s repetition and 90° excitation by optimizing both the Ernst angle and the repetition rate is moderate. For those molecules, it is therefore advocated to use a standard 90° excitation.

8. HMBC spectra recorded on isopropylidene glycerol: Ernst angle effect



Fig 7 HMBC spectra and extracted F_2 rows recorded on isopropylidene glycerol using the pulse sequence above. The spectra were recorded over 5 ppm (^{1}H) × 120 ppm (^{13}C) as 1K × 128 data points, 2 transients, corresponding to an acquisition time of 0.25 s. A recovery delay of 0.2 s was used, leading to a total experimental time of approximately 2,5 min. Data were zero-filled to 2K × 1K points. (a) recorded using a 90° excitation, and (b) recorded using the optimized Ernst angle excitation (130°) determined above. A gain of approximately 50% is realized for the last case using optimized Ernst angle. Note the very efficient $^{1}J_{CH}$ suppression and the absence of F_1 ridges.